
But de l’activité : élaborer un programme, gérer des tableaux de nombres

Compétences engagées :
✓ programmation de Xcas
✓ boucles « pour » imbriquées
✓ gestion de tableaux de nombres
Pré-requis :
✓ définition d’un carré magique
✓ somme des n premiers entiers naturels non nuls
✓ boucle « pour »

Matériels utilisés :
✓ un ordinateur équipé de Xcas
✓ une imprimante
✓ papier, crayon pour étude à la main
Durée indicative : 1 heure
Noms du logiciel utilisé : Xcas

Déroulement de la séance :
Définition, propriétés des carrés magiques.
Présentation des carrés magiques d’ordre impair : 3, 5, 7.
Méthode de Bachet pour obtenir une solution.
Présentation des commandes de Xcas pour la gestion des tableaux de nombres :
matrices, matrices par blocs, transposition
effet des commandes « makemat », « matrix », « tran », « blockmatrix »

Listing 1 – Carré magique d’ordre impair
// Procédure de Bachet de Méziriac pour obtenir un carré magique d’ordre\

n, n impair.
CarreMagique(n):={

if (est_pair(n)) {
return(n+"␣est␣pair.␣La␣méthode␣de␣Bachet␣de␣Méziriac␣nécessite␣un␣\

nombre␣entier␣impair.");
}
else {

// Déclaration des variables locales à la procédure
local M,l,d,C,N,E,S,O,Z;

// On crée une tableau ’M’, à (2n-1) lignes et (2n-1) colonnes , \
rempli de zéros.

M:=matrix (2*n-1,2*n-1,0);

// Deux boucles pour imbriquées (’l’ pour ligne et ’d’ pour \
diagonale),

// pour placer la liste des entiers de 1 à n^2 dans des diagonales \
descendantes

// de longueur n, à partir de la position (0,n-1).
// (la numérotation des lignes et colonnes se fait à partir de 0.)
for (l:=0;l<=n-1;l++){

for (d:=0;d<=n-1;d++){
M[0+d+l,(n-1)+d-l]:=1+d+l*n;

}

}

// Extraction du tableau central ’C’.
C:=M[(n-1) /2..(3*n-3)/2,(n-1) /2..(3*n-3) /2];
// ’N’, ’S’, ’E’, ’O’ sont les quatre tableaux exterieurs au tableau\

’C’,
// respectivement aux Nord , Sud , Est et Ouest du tableau ’C’
N:=M[0..(n-3)/2,(n-1) /2..(3*n-3) /2];
S:=M[(3*n-1) /2..2*n-2,(n-1) /2..(3*n-3) /2];
E:=M[(n-1) /2..(3*n-3) /2,(3*n-1) /2..2*n-2];
O:=M[(n-1) /2..(3*n-3) /2 ,0..(n-3) /2];

// On intègre les tableaux ’N’, ’S’, ’E’, ’O’ dans des tableaux
// à n lignes et n colonnes.
Z:=matrix((n+1)/2,n,0);
N:=blockmatrix(2,1,[Z,N]);
S:=blockmatrix(2,1,[S,Z]);

Z:=matrix(n,(n+1)/2,0);
E:=blockmatrix(1,2,[E,Z]);
O:=blockmatrix(1,2,[Z,O]);

// On retourne le tableau ’C’ contenant le carre magique d’ordre n,
// obtenu en assemblant les cinq tableaux précédents.
C:=C+N+S+E+O;
return C;

}
}:;

Variantes / Pour aller plus loin / Références :
✓ Le source fourni contient une petite amélioration qui n’est pas demandé dans la fiche élève : la procédure

de calcul du carré magique est englobée dans un test de parité de l’entier.
✓ La procédure échoue pour la valeur n=1. Cela est dû à la façon dont Xcas gère les matrices extraites

d’une autre pour un sous-ensemble d’indices vide : le fonctionnement n’est pas la même selon qu’on
travaille sur les lignes ou les colonnes. Ce n’est pas un bug ; voir le fil suivant sur le forum de Xcas :
http://pcm1.e.ujf-grenoble.fr/XCAS/viewtopic.php?f=3&t=588
On peut supposer qu’un élève sache générer à la main un carré magique de taille 1, mais on peut aussi
lui demander de corriger ce défaut à l’aide d’un test supplémentaire.

